

# DYNARE COURSE

## Amsterdam University

Introduction to Dynare

Michel Juillard

October 9, 2008

1. computes the solution of deterministic models (arbitrary accuracy)
2. computes first and second order approximation to solution of stochastic models
3. estimates (maximum likelihood or Bayesian approach) parameters of DSGE models
4. computes optimal policy.
5. performs global sensitivity analysis of a model (thanks to Marco Ratto's tools)



## The general problem

Deterministic, perfect foresight, case:

$$f(y_{t+1}, y_t, y_{t-1}, u_t) = 0$$

$y$  : vector of endogenous variables

$u$  : vector of exogenous shocks



## Solution of deterministic models

- ▶ based on work of Laffargue, Boucekkine and myself
- ▶ recently much accelerated by Mihoubi
- ▶ approximation: impose return to equilibrium in finite time instead of asymptotically
- ▶ computes the trajectory of the variables numerically
- ▶ uses a Newton-type method
- ▶ usefull to study full implications of non-linearities

Stochastic case:

$$E_t \{f(y_{t+1}, y_t, y_{t-1}, u_t)\} = 0$$

$y$  : vector of endogenous variables

$u$  : vector of exogenous shocks

$$E(u_t) = 0$$

$$E(u_t u_t') = \Sigma_u$$

$$E(u_t u_\tau') = 0 \quad t \neq \tau$$



## First order approximation

$$\begin{aligned} y_t &= \hat{g}(y_{t-1}, u_t) \\ &= \bar{y} + A\hat{y}_{t-1} + Bu_t \end{aligned}$$

with  $\hat{y}_t = y_t - \bar{y}$ .

Method proposed by Klein (2000) and Sims (2002).

DYNARE computes also theoretical moments and IRFs.



In a stochastic framework, the unknowns are the decision functions:

$$y_t = g(y_{t-1}, u_t)$$

For a large class of DSGE models, DYNARE computes approximated decision rules and transition equations by a perturbation method.



## Second order approximation

Two features:

- ▶ decision rules and transition functions are 2nd order polynomials
- ▶ departure from certainty equivalence: the variance of future shocks matters

Decision rules and transition equations of the form

$$y_t = \bar{y} + A\hat{y}_{t-1} + Bu_t + 0.5 (\hat{y}_{t-1}' C \hat{y}_{t-1} + u_t' D u_t) + \hat{y}_{t-1}' F u_t + \Delta(\Sigma_u)$$

Method suggested by K. Judd, developed by C. Sims (2002), S. Schmitt-Grohe and M. Uribe (2003), F. Collard and M. Juillard (2000).



## Estimation

DYNARE estimates the structural parameters of a model based on a linear approximation.

$$E_t \{f(y_{t+1}, y_t, y_{t-1}, u_t; \theta)\} = 0$$

### Estimation steps:

1. computes the steady state
2. linearizes the model
3. solves the linearized model
4. computes the log-likelihood via the Kalman filter
5. finds the maximum of the likelihood or posterior mode
6. simulates posterior distribution with Metropolis algorithm
7. computes various statistics on the basis of the posterior distribution
8. computes smoothed values of unobserved variables
9. computes forecasts and confidence intervals

## Optimal policy

$$\max_{z_t} \sum_{\tau=t}^{\infty} \beta^{\tau-t} U(y_{\tau}, z_{\tau}, u_{\tau})$$

s.t.

$$E_t \{f(y_{\tau+1}, y_{\tau}, y_{\tau-1}, z_{\tau}, u_{\tau})\} = 0$$

where  $U()$  is the objective function of the public authority and  $z_{\tau}$  a set of policy instruments.



## A simple example

```
var y inf r;
varexo e_y e_inf;

parameters delta sigma alpha kappa gamma1 gamma2;

delta = 0.44;
kappa = 0.18;
alpha = 0.48;
sigma = -0.06;
gamma1 = 1.5;
gamma2 = 0.5;
```



## A simple example (continued)

```
model(linear);
y = delta*y(-1)+(1-delta)*y(+1)
    +sigma*(r-inf(+1))+e_y;
inf = alpha*inf(-1)+(1-alpha)*inf(+1)
    +kappa*y+e_inf;
r = gamma1*inf+gamma2*y;
end;

shocks;
var e_y; stderr 0.63;
var e_inf; stderr 0.4;
end;

steady;
check;

stoch_simul;
```



- ▶ the Dynare toolbox solves generic problems
- ▶ the parser reads your \*.mod file and translates it in specific Matlab files
- ▶ *filename.m*: main Matlab script for your model
- ▶ *filename\_static.m*: static model
- ▶ *filename\_dynamic.m*: dynamic model

- ▶ Standard Cash-in-advance model (see also Nason and Cogley, 1994)
- ▶ Portfolio adjustment cost model
- ▶ Main Results:
  - ▶ Standard CIA model outperforms the PAC model in terms of both better in-sample properties and posterior distribution
  - ▶ The PAC model outperforms the CIA model when analyzing the response to a money growth shock.



## Standard CIA model

- ▶ Three agents: Household, a firm and a financial intermediary
- ▶ Decisions are made **after** the current period surprise change in money growth and technology.

## Firm

The firm chooses desired capital,  $K_t$ , labour demand,  $N_t$ , dividends  $F_t$  and loans  $L_t$

$$\max_{\{K_t, N_t, F_t\}_{t=0}^{\infty}} E_0 \left[ \sum_{t=0}^{\infty} \beta^t \frac{F_t}{C_{t+1} P_{t+1}} \right]$$

s.t.

$$\begin{aligned} F_t &\leq L_t + P_t Y_t - K_t + (1 - \delta) K_{t-1} - W_t N_t - R_{F,t} L_t \\ W_t N_t &\leq L_t \\ Y_t &= K_{t-1}^{\alpha} (A_t N_t)^{1-\alpha} \end{aligned}$$

where  $Y_t$  is output,  $A_t$ , technical change index,  $P_t$ , the price index,  $W_t$ , the wage rate, and  $R_{F,t}$ , the gross interest rate on loans.



The following maximisation problem:

$$\max_{\{B_t, L_t, D_t\}_{t=0}^{\infty}} E_0 \left[ \sum_{t=0}^{\infty} \beta^t \frac{B_t}{C_{t+1} P_{t+1}} \right]$$

s.t.

$$B_t = D_t + R_{F,t} L_t - R_{H,t} D_t - L_t + M_t - M_{t-1}$$

$$L_t \leq M_t - M_{t-1} + D_t$$

where  $B_t$  is dividends from the financial intermediaries,  $D_t$ , the deposits, and  $R_{H,t}$  is the gross deposit interest rate.



## Market clearing conditions

- ▶ Labor market

$$H_t = N_t$$

- ▶ Money market

$$P_t C_t = M_t$$

- ▶ Goods market

$$C_t + (K_t - (1 - \delta)K_{t-1}) = K_{t-1}^{\alpha} (A_t N_t)^{1-\alpha}$$



The household chooses consumption  $C_t$ , hours worked  $H_t$ , and (non-negative) deposits  $D_t$  so that

$$\max_{\{C_t, H_t, M_t, D_t\}_{t=0}^{\infty}} E_0 \left[ \sum_{t=0}^{\infty} \beta^t [(1 - \phi) \log C_t + \phi \log(1 - H_t)] \right]$$

s.t.

$$P_t C_t \leq M_{t-1} - D_t + W_t H_t$$

$$D_t \geq 0$$

$$M_t = M_{t-1} - D_t + W_t H_t - P_t C_t + R_{H,t} D_t + F_t + B_t$$



## Shock processes

Technology:

$$g_{A_t} = \frac{A_t}{A_{t-1}}$$

$$\log g_{A_t} = \gamma + \varepsilon_{A_t}$$

Money:

$$g_{M_t} = \frac{M_t}{M_{t-1}}$$

$$\log g_{M_t} = (1 - \rho) \log g_{M^*} + \rho \log g_{M_{t-1}} + \varepsilon_{M_t}$$



## Optimality conditions

$$E_t \left\{ \frac{P_t}{C_{t+1} P_{t+1}} \right\} = \beta E_t \left\{ \frac{P_{t+1} \alpha K_t^{\alpha-1} (A_{t+1} N_{t+1})^{1-\alpha} + 1 - \delta}{C_{t+2} P_{t+2}} \right\}$$

$$\frac{\psi}{1-\psi} \frac{C_t P_t}{1-N_t} = W_t$$

$$\frac{1}{C_t P_t} = \beta E_t \left\{ \frac{R_t}{C_{t+1} P_{t+1}} \right\}$$

$$R_t = \frac{P_t (1-\alpha) K_{t-1}^\alpha A_t^{1-\alpha} N_t^{-\alpha}}{W_t}$$



## Other equilibrium conditions

$$W_t = \frac{L_t}{N_t}$$

$$L_t = M_t - M_{t-1} + D_t$$

$$P_t C_t = M_t$$

$$C_t + (K_t - (1-\delta)K_{t-1}) = K_{t-1}^\alpha (A_t N_t)^{1-\alpha}$$

$$g_{A_t} = \frac{A_t}{A_{t-1}}$$

$$\log g_{A_t} = \gamma + \varepsilon_{A_t}$$

$$g_{M_t} = \frac{M_t}{M_{t-1}}$$

$$\log g_{M_t} = (1-\rho) \log g_{M_t}^* + \rho \log g_{M_{t-1}} + \varepsilon_{M_t}$$



## Stationarized model

In order for a local approximation to make sense, the model variables must fluctuate in the neighborhood of the steady state. It is therefore necessary to stationarize the model, before computing the local approximation of the solution.

- ▶  $K_t$ ,  $C_t$ , are detrended by the productivity  $A_t$ ,
- ▶  $P_t$  is detrended by  $M_{t-1}/A_t$
- ▶  $W_t$ ,  $L_t$ , and  $D_t$  are detrended by  $M_{t-1}$ .
- ▶  $R_t$  and  $N_t$  are stationary



## First-order conditions

Euler equation:

$$E_t \left\{ \frac{\hat{P}_t \frac{M_{t-1}}{A_t}}{\hat{C}_{t+1} A_{t+1} \hat{P}_{t+1} \frac{M_t}{A_{t+1}}} \right\}$$

$$= \beta E_t \left\{ P_{t+1} \frac{M_t}{A_{t+1}} \frac{\alpha (\hat{K}_t A_t)^{\alpha-1} (A_{t+1} N_{t+1})^{1-\alpha} + 1 - \delta}{\hat{C}_{t+2} A_{t+2} \hat{P}_{t+2} \frac{M_{t+1}}{A_{t+2}}} \right\}$$

after simplification

$$E_t \left\{ \frac{\hat{P}_t}{\hat{C}_{t+1} \hat{P}_{t+1} g_{M_t}} \right\} = \beta E_t \left\{ P_{t+1} \frac{\alpha g_{A_{t+1}}^{-\alpha} \hat{K}_t^{\alpha-1} N_{t+1}^{1-\alpha} + \frac{1-\delta}{g_{A_{t+1}}}}{\hat{C}_{t+2} \hat{P}_{t+2} g_{M_{t+1}}} \right\}$$



## First-order conditions (continued)

Firm's borrowing constraint:

$$\widehat{W}_t M_{t-1} = \frac{\widehat{L}_t M_{t-1}}{N_t}$$

after simplification

$$\widehat{W}_t = \frac{\widehat{L}_t}{N_t}$$

Intertemporal labor market optimality condition:

$$\frac{\psi}{1-\psi} \frac{\widehat{C}_t A_t \widehat{P}_t \frac{M_{t-1}}{A_t}}{1-N_t} = \widehat{W}_t M_{t-1}$$

after simplification

$$\frac{\psi}{1-\psi} \frac{\widehat{C}_t \widehat{P}_t}{1-N_t} = \widehat{W}_t$$

Navigation icons

## Other equilibrium conditions

$$\widehat{W}_t = \frac{\widehat{L}_t}{N_t}$$

$$\widehat{L}_t = g_{M_t} - 1 + \widehat{D}_t$$

$$\widehat{P}_t \widehat{C}_t = g_{M_t}$$

$$\widehat{C}_t + \widehat{K}_t - (1-\delta) \frac{\widehat{K}_{t-1}}{g_{A_t}} = \left( \frac{\widehat{K}_{t-1}}{g_{A_t}} \right)^\alpha N_t^{1-\alpha}$$

$$\log g_{A_t} = \gamma + \varepsilon_{A_t}$$

$$\log g_{M_t} = (1-\rho) \log g_{M^*} + \rho \log g_{M_{t-1}} + \varepsilon_{M_t}$$

Navigation icons

## First order conditions (continued)

Equilibrium interest rate:

$$R_t = \frac{(1-\alpha) \widehat{P}_t \frac{M_{t-1}}{A_t} \left( \widehat{K}_{t-1} A_{t-1} \right)^{\alpha-1} (A_t N_t)^{-\alpha}}{\widehat{W}_t M_{t-1}}$$

after simplification

$$R_t = \frac{(1-\alpha) \widehat{P}_t g_{A_t}^{-\alpha} \widehat{K}_{t-1}^{\alpha-1} N_t^{-\alpha}}{\widehat{W}_t}$$

Credit market optimality condition:

$$\frac{1}{\widehat{C}_t A_t \widehat{P}_t \frac{M_{t-1}}{A_t}} = \beta E_t \left\{ \frac{R_t}{\widehat{C}_{t+1} A_{t+1} \widehat{P}_{t+1} \frac{M_t}{A_{t+1}}} \right\}$$

after simplification

$$\frac{1}{\widehat{C}_t \widehat{P}_t} = \beta E_t \left\{ \frac{R_t}{g_{M_t} \widehat{C}_{t+1} \widehat{P}_{t+1}} \right\}$$

Navigation icons

## fs2000A.mod

```
var P C W R K D N L Y gA gM;
```

```
varexo e_a e_m;
```

```
parameters alp bet gam gMstar rho psi del;
```

```
alp = 0.33;
```

```
bet = 0.99;
```

```
gam = 0.003;
```

```
gMstar = 1.011;
```

```
rho = 0.7;
```

```
psi = 0.787;
```

```
del = 0.02;
```

Navigation icons

## fs2000A.mod (continued)

```
model;
P/(C(+1)*P(+1)*gM) = bet*P(+1)*(alp*gA(+1)^(-alp)*
    K^(alp-1)*N(+1)^(1-alp)+(1-del)/gA(+1))/
    (C(+2)*P(+2)*gM(+1));
W = L/N;
(psi/(1-psi))*(C*P/(1-N)) = W ;
R = P*(1-alp)*gA^(-alp)*K(-1)^alp*N^(-alp)/W;
1/(C*P) = bet*R/(gM*C(+1)*P(+1));
C+K = Y+(1-del)*K(-1)/gA;
P*C = gM;
gM-1+D = L;
Y = K(-1)^alp*N^(1-alp)*gA^(-alp);
log(gA) = gam + e_a;
log(gM) = (1-rho)*log(gMstar)+rho*log(gM(-1))+e_m;
end;
```

◀ ▶ ⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍 ↺

## fs2000A.mod (continued)

```
shocks;
var e_a; stderr 0.014;
var e_m; stderr 0.005;
end;

steady;
check;

stoch_simul(order=1);
```

◀ ▶ ⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍 ↺

## fs2000A.mod (continued)

```
initval;
K = 6;
P = 2.25;
C = 0.45;
W = 4;
R = 1.02;
D = 0.85;
N = 0.19;
L = 0.86;
Y = 0.6;
gA = exp(gam);
gM = dMstar;
end;
```

◀ ▶ ⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍 ↺

## Dating variables in Dynare

Dynare will automatically recognize predetermined and non-predetermined variables, but you must observe a few rules:

- ▶ period  $t$  variables are set during period  $t$  on the basis of the state of the system at period  $t - 1$  and shocks observed at the beginning of period  $t$ .
- ▶ therefore, stock variables must be on an end-of-period basis: investment of period  $t$  determines the capital stock at the end of period  $t$ .

◀ ▶ ⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍 ↺

## Log-linearization

- ▶ Taking a log-linear approximation of a model is equivalent to take a linear approximation of a model with respect to the logarithm of the variables.
- ▶ In practice, it is sufficient to replace all occurrences of variable  $X$  with  $\exp(LX)$  where  $LX = \log X$ .
- ▶ It is possible to make the substitution for some variables and not others. You wouldn't want to take a log approximation of a variable whose steady state value is negative ...
- ▶ There is no evidence that log-linearization is more accurate than simple linearization. In a growth model, it is often more natural to do a log-linearization.



## Simple example

```
var PIE Y RR;
varexo E_PIE E_Y;
parameters ALPHA BETA GAMMA PHI PIE_STAR;

ALPHA = .5;
BETA = .5;
GAMMA = .5;
PHI = .5;
PIE_STAR = 0;

model(linear);
PIE = ALPHA*PIE(+1)+(1-ALPHA)*PIE(-1)
      + BETA*Y + E_PIE;
Y = -GAMMA*RR + 0*Y(-1) + E_Y;
RR = PHI*(PIE-PIE_STAR);
end;
```



## Forecast of calibrated models

- ▶ the solution has the form of a VAR
- ▶ forecast is how to return to equilibrium
- ▶ *a priori* uncertainty about some parameters
- ▶ 2 types of graphs
  1. Type I
    - ▶ mean forecast
    - ▶ confidence interval for mean forecast (without uncertainty about futur shocks)
    - ▶ confidence interval for point forecast (with uncertainty about futur shocks)
  2. Type II
    - ▶ mean forecast
    - ▶ confidence interval with uncertainty about futur shocks (for prior mean value of the parameters)
    - ▶ confidence interval for point forecast (with uncertainty about futur shocks)



## Simple example continued

```
shocks;
var E_PIE;
stderr .1;
var E_Y;
stderr .1;
end;

histval;
Y(0) = 1;
PIE(0) = 1;
RR(0) = -1;
end;
```



## Simple example continued

```
priors;  
ALPHA,BETA_PDF,0.5,0.1;  
PIE_STAR,NORMAL_PDF,0,0.2;  
stderr E_PIE,INV_GAMMA_PDF,.1,0.001;  
stderr E_Y,INV_GAMMA_PDF,.1,0.001;  
end;  
  
prior_analysis(forecast=10);
```

